Hydrogen atom: radial wave function using Ladder operator

Masatsugu Sei Suzuki
Department of Physics, SUNY at Binghamton
(Date: 2-02-15)

Derivation of radial wave function of hydrogen atom can be discussed using the ladder operators. The radial Hamiltonian of the hydrogen atom is strikingly similar to that of the three-dimensional simple harmonic oscillator. We use essentially the same technique, defining the dimensionless ladder operator (see the detail in Binney and Skinner).

---

1. Ladder operator

The Hamiltonian of hydrogen atom is given in terms of the radial momentum p_r and the total orbital angular momentum $\boldsymbol{L}^2$ as

$$\begin{aligned} H_l &= \frac{1}{2\mu}\boldsymbol{p}^2 - \frac{Ze^2}{r} \\ &= \frac{1}{2\mu}(p_r{}^2 + \frac{\boldsymbol{L}^2}{r^2}) - \frac{Ze^2}{r} \\ &= \frac{p_r{}^2}{2\mu} + \frac{l(l+1)\hbar^2}{2\mu r^2} - \frac{Ze^2}{r} \\ &= \frac{\hbar^2}{2\mu}[\frac{p_r{}^2}{\hbar^2} + \frac{l(l+1)}{r^2} - \frac{2Z}{ar}] \end{aligned},$$

where the radial momentum operator p_r is given by

$$p_r = -i\hbar\frac{1}{r}\frac{\partial}{\partial r}r = -i\hbar(\frac{\partial}{\partial r} + \frac{1}{r}).$$

and a is defined by

$$a = \frac{\hbar^2}{\mu e^2}.$$

When $\mu = m$, a is the Bohr radius a_B. H_l is the Hamiltonian for a particle that moves in one dimension in the effective potential

$$V_{eff}(r) = \frac{l(l+1)\hbar^2}{2\mu r^2} - \frac{Ze^2}{r} = \frac{\hbar^2}{2\mu}[\frac{l(l+1)}{r^2} - \frac{2Z}{ar}].$$

Here we defined the raising operator;

$$A_l = \frac{a}{\sqrt{2}}[\frac{ip_r}{\hbar} - \frac{l+1}{r} + \frac{Z}{(l+1)a}],$$

The lowering operator is also defined as

$$A_l^+ = \frac{a}{\sqrt{2}}[-\frac{ip_r}{\hbar} - \frac{l+1}{r} + \frac{Z}{(l+1)a}].$$

We now calculate $A_l^+ A_l$

$$\begin{aligned}
A_l^+ A_l &= \frac{a}{\sqrt{2}}[-\frac{ip_r}{\hbar} - \frac{l+1}{r} + \frac{Z}{(l+1)a}]\frac{a}{\sqrt{2}}[\frac{ip_r}{\hbar} - \frac{l+1}{r} + \frac{Z}{(l+1)a}] \\
&= \frac{a^2}{2}\{\frac{p_r^2}{\hbar^2} + (\frac{Z}{(l+1)a} - \frac{l+1}{r})^2 + \frac{i}{\hbar}[p_r, \frac{l+1}{r}]\} \\
&= \frac{a^2}{2}\{\frac{p_r^2}{\hbar^2} + \frac{Z^2}{(l+1)^2 a^2} + \frac{(l+1)^2}{r^2} - \frac{2Z}{ar} + \frac{i}{\hbar}[p_r, \frac{l+1}{r}]\} \\
&= \frac{a^2}{2}\{\frac{p_r^2}{\hbar^2} + \frac{Z^2}{(l+1)^2 a^2} + \frac{(l+1)^2}{r^2} - \frac{2Z}{ar} - \frac{l+1}{r^2}\} \\
&= \frac{a^2}{2}\{\frac{p_r^2}{\hbar^2} + \frac{l(l+1)}{r^2} - \frac{2Z}{ar} + \frac{Z^2}{(l+1)^2 a^2}\} \\
&= \frac{\mu a^2}{\hbar^2} H_l + \frac{Z^2}{2(l+1)^2}
\end{aligned}$$

leading to the expression of H_l as

$$H_l = \frac{\hbar^2}{\mu a^2}[A_l^+ A_l - \frac{Z^2}{2(l+1)^2}],$$

where

$$\left[p_r, \frac{l+1}{r}\right]\psi(r) = p_r \frac{l+1}{r}\psi - \frac{l+1}{r} p_r\psi$$

$$= \frac{\hbar}{i}\frac{1}{r}\frac{\partial}{\partial r}(r\frac{l+1}{r}\psi) - \frac{l+1}{r}\frac{\hbar}{i}\frac{1}{r}\frac{\partial}{\partial r}(r\psi)$$

$$= \frac{\hbar}{i}\frac{(l+1)}{r}\frac{\partial}{\partial r}\psi - \frac{l+1}{r}\frac{\hbar}{i}\frac{1}{r}(\psi + r\frac{\partial}{\partial r}\psi)$$

$$= -\frac{l+1}{r^2}\frac{\hbar}{i}\psi$$

The commutation relation:

$$[A_l^+, A_l] = \frac{a^2}{2}[-\frac{ip_r}{\hbar} - \frac{l+1}{r} + \frac{Z}{(l+1)a}, \frac{ip_r}{\hbar} - \frac{l+1}{r} + \frac{Z}{(l+1)a}]$$

$$= a^2 \frac{i}{\hbar}[p_r, \frac{l+1}{r}]$$

$$= -\frac{a^2(l+1)}{r^2}$$

$$= -\frac{\mu a^2}{\hbar^2}(H_{l+1} - H_l)$$

or

$$[A_l, A_l^+] = \frac{\mu a^2}{\hbar^2}(H_{l+1} - H_l),$$

since

$$H_{l+1} - H_l = \frac{\hbar^2}{2\mu r^2}[(l+2)(l+1) - l(l+1)] = \frac{\hbar^2}{\mu r^2}(l+1)].$$

Using the above commutation relation, we get

$$
\begin{aligned}
H_l &= \frac{\hbar^2}{\mu a^2}[A_l^+ A_l - \frac{Z^2}{2(l+1)^2}] \\
&= \frac{\hbar^2}{\mu a^2}\{[A_l^+, A_l] + A_l A_l^+ - \frac{Z^2}{2(l+1)^2}] \\
&= \frac{\hbar^2}{\mu a^2}\{-\frac{a^2\mu}{\hbar^2}(H_{l+1} - H_l) + A_l A_l^+ - \frac{Z^2}{2(l+1)^2}] \\
&= -(H_{l+1} - H_l) + \frac{\hbar^2}{\mu a^2}[A_l A_l^+ - \frac{Z^2}{2(l+1)^2}]
\end{aligned}
$$

or

$$
H_{l+1} = \frac{\hbar^2}{\mu a^2}[A_l A_l^+ - \frac{Z^2}{2(l+1)^2}].
$$

((**Note**))

$$
[A_l^+, A_{l+1}^+] = -\frac{a^2}{2r^2},
$$

$$
[A_l, A_{l+1}] = \frac{a^2}{2r^2},
$$

$$
[A_l, A_{l+1}^+] = \frac{a^2}{2r^2}(2l+3),
$$

$$
[A_l^+, A_{l+1}] = -\frac{a^2}{2r^2}(2l+3).
$$

We also note that

$$[A_l^+, H_l] = \frac{\hbar^2}{\mu a^2}[A_l^+, A_l^+ A_l]$$

$$= \frac{\hbar^2}{\mu a^2} A_l^+ [A_l^+, A_l]$$

$$= -A_l^+ (H_{l+1} - H_l)$$

leading to

$$H_l A_l^+ = A_l^+ H_{l+1}.$$

or

$$A_l H_l = H_{l+1} A_l.$$

Similarly

$$[A_l, H_l] = \frac{\hbar^2}{\mu a^2}[A_l, A_l^+ A_l]$$

$$= \frac{\hbar^2}{\mu a^2}[A_l, A_l^+] A_l$$

$$= (H_{l+1} - H_l) A_l$$

or

$$H_{l+1} A_l = A_l H_l.$$

2. **Eigenvalue problem**

Suppose that

$$H_l |E, l\rangle = E |E, l\rangle, \qquad H_{l+1} |E, l+1\rangle = E |E, l+1\rangle.$$

Then we get

$$H_{l+1} A_l |E, l\rangle = A_l H_l |E, l\rangle = E A_l |E, l\rangle.$$

This means that $A_l|E,l\rangle$ is the eigenket of H_{l+1} with the eigenvalue *E*. Then we have

$$A_l|E,l\rangle \propto |E,l+1\rangle.$$

The operator A_l creates a stationary state with the same energy *E* for the one step up from *l* to (*l*+1). We also get

$$H_l A_l^+|E,l+1\rangle = A_l^+ H_{l+1}|E,l+1\rangle = EA_l^+|E,l+1\rangle.$$

This means that $A_l^+|E,l+1\rangle$ is the eigenket of H_l with the eigenvalue *E*. Then we have

$$A_l^+|E,l+1\rangle \propto |E,l\rangle.$$

The operator A_l^+ creates a stationary state with the same energy *E* for the one step down from (*l*+1) to *l*,

.

The kinetic energy is shifted from $\dfrac{p_r^{\ 2}}{2\mu}$ to $\dfrac{l(l+1)\hbar^2}{2\mu r^2}$ as *l* increases up to $l = l_{max}$. Here we assume that

$$A_{l\max}|E,l_{\max}\rangle = 0,$$

or

$$\langle E,l_{\max}|A_{l\max}^{\ +}A_{l_{\max}}|E,l_{\max}\rangle = 0.$$

Since

$$A_{l\max}^{\ +}A_{l\max} = \frac{\mu a^2}{\hbar^2}H_{l_{\max}} + \frac{Z^2}{2(l_{\max}+1)^2},$$

we have

$$\langle E, l_{\max} | \frac{\mu a^2}{\hbar^2} H_{l_{\max}} + \frac{Z^2}{2(l_{\max}+1)^2} | E, l_{\max} \rangle = 0 \,,$$

or

$$\langle E, l_{\max} | \frac{\mu a^2}{\hbar^2} H_{l\max} | E, l_{\max} \rangle = -\frac{Z^2}{2(l_{\max}+1)^2} \,.$$

Noting that

$$H_{l\max} | E, l_{\max} \rangle = E | E, l_{\max} \rangle \,,$$

we get

$$\begin{aligned} \langle E, l_{\max} | H_{l_{\max}} | E, l_{\max} \rangle = E &= -\frac{\hbar^2}{2\mu a^2} \frac{Z^2}{(l_{\max}+1)^2} \\ &= -\frac{\hbar^2}{2\left(\frac{\hbar^2}{\mu e^2}\right)^2 \mu} \frac{Z^2}{n^2} \qquad , \\ &= -\frac{\mu Z^2 e^4}{2n^2\hbar^2} \end{aligned}$$

which agrees well with the prediction from the Bohr model. Here we note that n is the principal quantum number and is given by

$$n = l_{\max} + 1 \,,$$

The energy eigenvalue is

$$E = E_n = -\frac{\mu e^4 Z^2}{2n^2\hbar^2} \,.$$

3. Energy eigenfunction $\langle r | E, l_{\max} \rangle$

$$A_{l_{\max}}\left|E,l_{\max}\right\rangle = 0\,,$$

or

$$A_{l_{\max}}\left|E,l_{\max}\right\rangle = \frac{a}{\sqrt{2}}[\frac{ip_r}{\hbar} - \frac{l_{\max}+1}{r} + \frac{Z}{(l_{\max}+1)a}]\left|E,l_{\max}\right\rangle = 0\,.$$

We note that

$$\left\langle r\right|(\frac{ip_r}{\hbar} - \frac{n}{r} + \frac{Z}{na})\left|E,l_{\max}\right\rangle = 0\,,$$

where

$$p_r = -i\hbar\frac{1}{r}\frac{\partial}{\partial r}r = -i\hbar(\frac{\partial}{\partial r} + \frac{1}{r})\,.$$

Then we have the differential equation,

$$\frac{d}{dr}\chi(r) - \frac{n-1}{r}\chi(r) + \frac{Z}{na}\chi(r) = 0\,,$$

with

$$\chi(r) = \left\langle r\middle|E,l_{\max}\right\rangle.$$

The solution of this differential equation is given by

$$\chi_{l_{\max}}(r) = \left\langle r\middle|E,l_{\max}\right\rangle = Cr^{n-1}\exp(-\frac{Zr}{na})\,,$$

where

$$C = \frac{2^n}{\sqrt{n(2n-1)!}}\left(\frac{Z}{na}\right)^{n-1+\frac{3}{2}} = \frac{2^n}{\sqrt{n(2n-1)!}}\left(\frac{Z}{na}\right)^{\frac{3}{2}}\left(\frac{Z}{na}\right)^{n-1}.$$

Using this constant, we get the normalized wave function

$$\chi_{l_{\max}}(r) = \langle r | E, l_{\max} \rangle = \frac{2^n}{\sqrt{n(2n-1)!}} \left(\frac{Z}{na}\right)^{\frac{3}{2}} \left(\frac{Zr}{na}\right)^{n-1} \exp(-\frac{Zr}{na})$$

$$= \frac{2^n}{\sqrt{n(2n-1)!}} \frac{1}{n^{n-1}} \left(\frac{Z}{na}\right)^{\frac{3}{2}} \left(\frac{Zr}{a}\right)^{n-1} \exp(-\frac{Zr}{na})$$

$$= \frac{2^n \sqrt{2}}{\sqrt{(2n)!}} \frac{1}{n^{n-1}} \left(\frac{Z}{na}\right)^{\frac{3}{2}} \left(\frac{Zr}{a}\right)^{n-1} \exp(-\frac{Zr}{na})$$

We calculate the expectation value $\langle r \rangle$ and $\langle r^2 \rangle$.

$$\langle r \rangle = \int_0^\infty r^3 dr [\chi_{l_{\max}}(r)]^2 = \frac{a}{Z} n(n+\frac{1}{2}),$$

and

$$\langle r^2 \rangle = \int_0^\infty r^4 dr [\chi_{l_{\max}}(r)]^2 = \frac{a^2}{Z^2} n^2 (n+1)(n+\frac{1}{2}).$$

Then we have the ratio defined by

$$\frac{\langle r^2 \rangle}{\langle r \rangle^2} = \frac{n+1}{n+\frac{1}{2}} = 1 + \frac{1}{2n+1}.$$

The uncertainty in *r* is

$$\sqrt{\langle r^2 \rangle - \langle r \rangle^2} = \frac{\langle r \rangle}{\sqrt{2n+1}} = \frac{a}{\sqrt{2}Z} n \sqrt{n+\frac{1}{2}} \propto n^{3/2}.$$

As *n* increases, the uncertainty in *r* increases as $n^{3/2}$.

((**Mathematica**))

```
Clear["Global`*"];
```

```
eq1 = χ'[r] - (n - 1)/r χ[r] + Z/(n a) χ[r] == 0
```

$$-\frac{(-1+n)\,\chi[r]}{r}+\frac{Z\,\chi[r]}{a\,n}+\chi'[r]==0$$

```
eq2 = DSolve[eq1, χ[r], r] // Simplify
```

$$\left\{\left\{\chi[r]\to e^{-\frac{r Z}{a n}}\, r^{-1+n}\, C[1]\right\}\right\}$$

```
χ[r_] = χ[r] /. eq2[[1]]
```

$$e^{-\frac{r Z}{a n}}\, r^{-1+n}\, C[1]$$

```
k1 = Integrate[χ[r]^2 r^2, {r, 0, ∞}] //
  Simplify[#, {n > 0, n ∈ Integers, Z > 0,
     a > 0}] &
```

$$4^{-n}\, n \left(\frac{Z}{a\, n}\right)^{-1-2 n} C[1]^2\, \text{Gamma}[2 n]$$

```
eq3 = Solve[k1 == 1, C[1]]
```

$$\left\{\left\{C[1] \to -\frac{2^n \left(\frac{Z}{a\, n}\right)^{\frac{1}{2}(1+2 n)}}{\sqrt{n}\, \sqrt{\text{Gamma}[2 n]}}\right\}, \left\{C[1] \to \frac{2^n \left(\frac{Z}{a\, n}\right)^{\frac{1}{2}(1+2 n)}}{\sqrt{n}\, \sqrt{\text{Gamma}[2 n]}}\right\}\right\}$$

```
h1 = C[1] /. eq3[[2]]
```

$$\frac{2^n \left(\frac{Z}{a\, n}\right)^{\frac{1}{2}(1+2 n)}}{\sqrt{n}\, \sqrt{\text{Gamma}[2 n]}}$$

4. Recursion relation

We assume that

$$A_l^{+}\left|E, l+1\right\rangle = c_l \left|E, l\right\rangle .$$

Using the formula

$$H_l = \frac{\hbar^2}{\mu a^2}[A_l^{+} A_l - \frac{Z^2}{2(l+1)^2}], \qquad H_{l+1} = \frac{\hbar^2}{\mu a^2}[A_l A_l^{+} - \frac{Z^2}{2(l+1)^2}]$$

we get

$$\left\langle E, l+1 \right| A_l A_l^{+} \left| E, l+1 \right\rangle = \left|c_l\right|^2 .$$

Using the relation

$$A_l A_l^+ = \frac{\mu a^2}{\hbar^2} H_{l+1} + \frac{Z^2}{2(l+1)^2},$$

we get

$$\langle E, l+1 | \frac{\mu a^2}{\hbar^2} H_{l+1} + \frac{Z^2}{2(l+1)^2} | E, l+1 \rangle = |c_l|^2,$$

or

$$\begin{aligned} |c_l|^2 &= \frac{a^2 \mu}{\hbar^2} E + \frac{Z^2}{2(l+1)^2} \\ &= -\frac{Z^2}{2(l_{\max}+1)^2} + \frac{Z^2}{2(l+1)^2} \\ &= \frac{Z^2}{2}[-\frac{1}{n^2} + \frac{1}{(l+1)^2}] \\ &= \frac{Z^2}{2}[\frac{n^2 - (l+1)^2}{n^2 (l+1)^2}] \\ &= \frac{Z^2}{2}[\frac{(n+l+1)(n-l-1)}{n^2 (l+1)^2}] \end{aligned},$$

with

$$E = -\frac{\hbar^2}{2\mu a^2} \frac{Z^2}{(l_{\max}+1)^2} = -\frac{\hbar^2 Z^2}{2\mu a^2 n^2}.$$

By the appropriate choice of the phase factor for c_l, we have

$$c_l = \frac{1}{f(n,l)} = -\frac{Z}{\sqrt{2}} \sqrt{-\frac{1}{(l_{\max}+1)^2} + \frac{1}{(l+1)^2}} = -\frac{Z}{\sqrt{2}} \frac{\sqrt{(n+l+1)(n-l-1)}}{n(l+1)}.$$

Then we get

$$A_l \left| E, l+1 \right\rangle = \frac{1}{f(n,l)} \left| E, l \right\rangle ,$$

or

$$\begin{aligned}\psi(l) = \langle r | E, l \rangle &= f(n,l) \langle r | A_l | E, l+1 \rangle \\ &= f(n,l) A_l \psi(l+1) \\ &= B(n,l) \psi(l+1)\end{aligned}$$

where the operator $B(n,l)$ is defined by

$$B(n,l) = f(n,l) A_l .$$

and the operator A_l is

$$A_l = \frac{a}{\sqrt{2}} [\frac{1}{r} \frac{\partial}{\partial r} r - \frac{l+1}{r} + \frac{Z}{(l+1)a}]$$

Using the above recursion relation, we get

$$\psi(l = n-2) = B(n, l = n-2) \psi(l = n-1) ,$$

$$\begin{aligned}\psi(l = n-3) &= B(n, l = n-3) \psi(l = n-2) \\ &= B(n, l = n-3) B(n, l = n-2) \psi(l = n-1)\end{aligned}$$

$$\begin{aligned}\psi(l = n-4) &= B(n, l = n-4) \psi(l = n-3) \\ &= B(n, l = n-4) B(n, l = n-3) B(n, l = n-2) \psi(l = n-1)\end{aligned}$$

$$\begin{aligned}\psi(l = n-5) &= B(n, l = n-5) \psi(l = n-4) \\ &= B(n, l = n-5) B(n, l = n-4) B(n, l = n-3) B(n, l = n-2) \psi(l = n-1)\end{aligned}$$

and so on.

Using the Mathematica, the radial eigenfunctions are obtained as follows.

$n = 1$

$$R_{1,0}(r) = 2e^{-rZ/a}(\frac{Z}{a})^{3/2} \qquad n = 1,\ l = 0.$$

$n = 2$

$$R_{2,0}(r) = \frac{1}{\sqrt{2}}e^{-rZ/2a}(\frac{Z}{a})^{3/2}(1 - \frac{rZ}{2a}) \qquad n = 2,\ l = 0.$$

$$R_{2,1}(r) = \frac{1}{2\sqrt{6}}e^{-rZ/2a}(\frac{Z}{a})^{3/2}\frac{rZ}{a} \qquad n = 2,\ l = 1.$$

$n = 3$

$$R_{3,0}(r) = \frac{2}{3\sqrt{3}}e^{-rZ/3a}(\frac{Z}{a})^{3/2}(1 - \frac{2rZ}{3a} + \frac{2r^2Z^2}{27a^2}) \qquad n = 3,\ l = 0.$$

$$R_{3,1}(r) = \frac{4}{27}\sqrt{\frac{2}{3}}e^{-rZ/3a}(\frac{Z}{a})^{3/2}\frac{rZ}{a}(1 - \frac{rZ}{6a}) \qquad n = 3,\ l = 1.$$

$$R_{3,2}(r) = \frac{2}{81}\sqrt{\frac{2}{15}}e^{-rZ/3a}(\frac{Z}{a})^{3/2}\frac{r^2Z^2}{a^2} \qquad n = 3,\ l = 2.$$

((Mathematica)) **Recursion relation**

```
Clear["Global`*"]; Lmax = n - 1;
```

$$\texttt{f1}[n1\_, L1\_] := -\frac{n1\,(L1+1)}{\sqrt{(n1+L1+1)\,(n1-L1-1)}}\,\frac{\sqrt{2}}{Z};$$

The raising operator: A+

$$\texttt{AU}[n1\_, k1\_] := \texttt{f1}[n1, k1]\,\frac{a}{\sqrt{2}}\left(\frac{-1}{r}\,\texttt{D}[r\,\#, r] - \frac{k1+1}{r}\,\# + \frac{Z}{(k1+1)\,a}\,\#\right)\,\&;$$

$$\texttt{u}[r\_, L1max\_] := e^{-\frac{r Z}{a n}}\, r^{-1+n}\,\frac{2^n\left(\frac{Z}{a n}\right)^{\frac{1}{2}(1+2n)}}{\sqrt{n}\,\sqrt{(2n-1)!}}\ \texttt{/.}\ n \to L1max + 1\ \texttt{//}\ \texttt{Simplify}$$

n = 5; *l* = 4,3, 2, 1, 0

```
n = 5; Lmax = 4;
```

```
u[r, Lmax] // Simplify
```

$$\frac{2\sqrt{\frac{2}{35}}\,e^{-\frac{r Z}{5 a}}\,r^4\left(\frac{Z}{a}\right)^{11/2}}{140625}$$

```
AU[n, n - 2][u[r, Lmax]] // Simplify
```

$$\frac{2\sqrt{\frac{2}{35}}\, e^{-\frac{r Z}{5 a}}\, r^3 Z^4 \sqrt{\frac{Z}{a}}\, (20 a - r Z)}{46875\, a^5}$$

```
AU[n, n - 3][AU[n, n - 2][u[r, Lmax]]] // Simplify
```

$$\frac{2\sqrt{\frac{2}{35}}\, e^{-\frac{r Z}{5 a}}\, r^2 Z^3 \sqrt{\frac{Z}{a}}\, \left(525 a^2 - 70 a r Z + 2 r^2 Z^2\right)}{46875\, a^5}$$

```
AU[n, n - 3][AU[n, n - 3][AU[n, n - 2][u[r, Lmax]]]] //
 Simplify
```

$$\frac{1}{46875\, a^3}\sqrt{\frac{2}{35}}\, e^{-\frac{r Z}{5 a}}\, r \left(\frac{Z}{a}\right)^{5/2} \left(23625 a^3 - 5775 a^2 r Z + 400 a r^2 Z^2 - 8 r^3 Z^3\right)$$

```
AU[n, n - 4][
  AU[n, n - 3][
   AU[n, n - 3][AU[n, n - 2][u[r, Lmax]]]]] //
 Simplify
```

$$\frac{1}{328125\, a^4} \sqrt{\frac{2}{15}}\, e^{-\frac{r Z}{5 a}} \left(\frac{Z}{a}\right)^{3/2} \left(945000\, a^4 - 454125\, a^3 r Z + 64425\, a^2 r^2 Z^2 - 3360\, a\, r^3 Z^3 + 56\, r^4 Z^4\right)$$

n = 4; *l* = 3, 2, 1, 0

```
n = 4; Lmax = 3;
```

```
u[r, Lmax] // Simplify
```

$$\frac{e^{-\frac{r Z}{4 a}}\, r^3 \left(\frac{Z}{a}\right)^{9/2}}{768 \sqrt{35}}$$

```
AU[n, n - 2][u[r, Lmax]] // Simplify
```

$$\frac{e^{-\frac{r Z}{4 a}}\, r^2 Z^3 \sqrt{\frac{Z}{a}}\, (12\, a - r Z)}{768 \sqrt{5}\, a^4}$$

```
AU[n, n - 3][AU[n, n - 2][u[r, Lmax]]] // Simplify
```

$$\frac{e^{-\frac{r Z}{4 a}}\, r \left(\frac{Z}{a}\right)^{5/2} \left(80\, a^2 - 20\, a\, r Z + r^2 Z^2\right)}{256 \sqrt{15}\, a^2}$$

```
AU[n, n - 4][AU[n, n - 3][AU[n, n - 2][u[r, Lmax]]]] //
 Simplify
```

$$\frac{e^{-\frac{r Z}{4 a}} \left(\frac{Z}{a}\right)^{3/2} \left(192 a^3 - 144 a^2 r Z + 24 a r^2 Z^2 - r^3 Z^3\right)}{768 a^3}$$

n = 3; *l* = 2, 1, 0

```
n = 3; Lmax = 2;
```

```
u[r, Lmax] // Simplify
```

$$\frac{2}{81} \sqrt{\frac{2}{15}}\, e^{-\frac{r Z}{3 a}} r^2 \left(\frac{Z}{a}\right)^{7/2}$$

```
AU[n, n - 2][u[r, Lmax]] // FullSimplify
```

$$-\frac{2 \sqrt{\frac{2}{3}}\, e^{-\frac{r Z}{3 a}} r \left(\frac{Z}{a}\right)^{5/2} (-6 a + r Z)}{81 a}$$

```
AU[n, n - 3][AU[n, n - 2][u[r, Lmax]]] // Simplify
```

$$\frac{2 e^{-\frac{r Z}{3 a}} \left(\frac{Z}{a}\right)^{3/2} \left(27 a^2 - 18 a r Z + 2 r^2 Z^2\right)}{81 \sqrt{3}\, a^2}$$

n = 2; *l* =1, 0

```
n = 2; Lmax = 1;
u[r, Lmax] // Simplify
```

$$\frac{e^{-\frac{r Z}{2 a}} r \left(\frac{Z}{a}\right)^{5/2}}{2 \sqrt{6}}$$

```
AU[n, n - 2][u[r, Lmax]] // Simplify
```

$$\frac{e^{-\frac{r Z}{2 a}} \left(\frac{Z}{a}\right)^{3/2} (2 a - r Z)}{2 \sqrt{2} a}$$

n = 1; *l* = 0

```
n = 1; Lmax = 0;
u[r, Lmax] // Simplify
```

$$2 e^{-\frac{r Z}{a}} \left(\frac{Z}{a}\right)^{3/2}$$

REFERENCES

J. Binney and D. Skinner, *The Physics of Quantum Mechanics* (Oxford, 2014).

APPENDIX

Mathematica: Commutation relations

```
Clear["Global`*"];
```

The operator: A=AD

$$AD[L\_] := \frac{a}{\sqrt{2}}\left(\frac{1}{r}\,D[r\,\#,\,r] - \frac{L+1}{r}\,\# + \frac{Z}{(L+1)\,a}\,\#\right)\&;$$

The operator: A+=AU

$$AU[L\_] := \frac{a}{\sqrt{2}}\left(-\frac{1}{r}\,D[r\,\#,\,r] - \frac{L+1}{r}\,\# + \frac{Z}{(L+1)\,a}\,\#\right)\&;$$

The quantum mechanical operator

$$pr := -i\,\frac{\hbar}{r}\,D[r\,\#,\,r]\,\&;$$

The Hamiltonian of hydrogen atom. μ is a reduced mass.

$$H[L\_] := \left(\frac{1}{2\,\mu}\,pr[pr[\#]] + \frac{L\,(L+1)\,\hbar^2}{2\,\mu\,r^2}\,\# - \frac{Z\,e1^2}{r}\,\#\right)\&;$$

Al+ Al calculation

```
f1 = ħ^2/(μ a^2) (AU[L][AD[L][χ[r]]] - Z^2 χ[r]/(2 (L + 1)^2)) //
    Simplify;

f2 = H[L][χ[r]] // Simplify;

eq3 = (f1 - f2) /. {a → ħ^2/(μ e1^2)} // Simplify

0
```

The commutation relation
A[L] A+[L] - A+[L]A[L]

```
f3 =
 (AU[L][AD[L][χ[r]]] -  AD[L][AU[L][χ[r]]]) //
  Simplify
```

$$-\frac{a^2 (1+L)\,\chi[r]}{r^2}$$

```
f4 =
  -(a^2 μ/ħ^2) (H[L + 1][χ[r]] - H[L][χ[r]] // Simplify);
f3 - f4 // Simplify
```

0

Some comment on various kinds of commutation relations

A+[L] A+[L+1] - A+[L+1]+A[L]

```
AU[L][AU[L + 1][χ[r]]] -
  AU[L + 1][AU[L][χ[r]]] // Simplify
```

$$-\frac{a^2 \chi[r]}{2 r^2}$$

A[L] A[L+1] - A[L+1]A[L]

```
AD[L][AD[L + 1][χ[r]]] -
  AD[L + 1][AD[L][χ[r]]] // Simplify
```

$$\frac{a^2 \chi[r]}{2 r^2}$$

A[L] A+[L+1] - A+[L+1]A[L]

```
AD[L][AU[L + 1][χ[r]]] -
  AU[L + 1][AD[L][χ[r]]] // Simplify
```

$$\frac{a^2\ (3 + 2\,L)\ \chi[r]}{2\,r^2}$$

A+[L] A[L+1] - A[L+1]A+[L]

```
AU[L][AD[L + 1][χ[r]]] -
  AD[L + 1][AU[L][χ[r]]] // Simplify
```

$$-\frac{a^2\ (3 + 2\,L)\ \chi[r]}{2\,r^2}$$

H[L+1] - H[L]

$$\text{seq1} = \frac{\mu\,a^2}{\hbar^2}\ (H[L + 1][\chi[r]] - H[L][\chi[r]])\ //\ \text{Simplify}$$

$$\frac{a^2\,(1+L)\,\chi[r]}{r^2}$$

A[L] H[L] - H[L+1] A[L]=0

```
AD[L][H[L][χ[r]]] - H[L + 1][AD[L][χ[r]]] /.
   {a → ℏ^2/(μ e1^2)} // Simplify
```

0

H[L] AU[L] - AU[L] H[L+1]=0

```
H[L][AU[L][χ[r]]] - AU[L][H[L + 1][χ[r]]] /.
    {a → ℏ^2/(μ e1^2)} // Simplify // Simplify
```

0